

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
				Monarch2090
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	Warteschleife setzt auf 0
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife (vorher 0, nachher 0)
		#define	ZAHL09 r9	Zufallszahl über Eingänge 1-4
		#define	TASTE r9	Rote Taste zum Nachstart
		#define	STATUS rA	Wenn Sonderspiele, dann 2, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
0x01	F4C		DISP #4,GELD_PF_EINER	
0x02	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x03	F04		DZHX	
0x04	F4C	start	DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0x05	1E5		MOVI #E,DELTA_D	-20 Pf
0x06	BE2		CALL DeltaSub	
0x07	9FF		CMPI #F,GELD_DM_ZEHNER	Münzspeicher unter 0?
0x08	E00		BRZ start0	dann zurück zum Anfang
0x09	BEB		CALL Warteschleife	
0x0A	FEA		DOT STATUS	Keine Sonderspiele?
0x0B	E10		BRZ StartWalzen	dann direkt Spiel starten
0x0C	F0E		EXRM	
0x0D	F2D		DISP #2,S_EINER	Sonderspiele-Zähler anzeigen
0x0E	BEB		CALL Warteschleife	
0x0F	F0E		EXRM	
0x10	F50	StartWalzen	DISP #5,RO	Walzen darstellen
0x11	BF2	stopL	CALL HoleZahl	Linke Walze ermitteln
0x12	1F4		MOVI #F,LO	König
0x13	1D3		MOVI #D,LU	2 DM
0x14	989		CMPI #8,ZAHL09	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x15	D2A		BRC L_fertig	
0x16	1E4		MOVI #E,LO	Krone
0x17	193		MOVI #9,LU	1 DM
0x18	979		CMPI #7,ZAHL09	
0x19	D2A		BRC L_fertig	
0x1A	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x1B	969		CMPI #6,ZAHL09	
0x1C	D2A		BRC L_fertig	
0x1D	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x1E	959		CMPI #5,ZAHL09	
0x1F	D2A		BRC L_fertig	
0x20	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x21	949		CMPI #4,ZAHL09	
0x22	D2A		BRC L_fertig	
0x23	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x24	939		CMPI #3,ZAHL09	
0x25	D2A		BRC L_fertig	
0x26	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x27	919		CMPI #1,ZAHL09	
0x28	D2A		BRC L_fertig	
0x29	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x2A	145	L_fertig	MOVI #4,DELTA_D	Ausgang 3 auf high
0x2B	FE5		DOT DELTA_D	Generator stop, Zähler auf 0, Taster aktiv
0x2C	BEB		CALL Warteschleife	
0x2D	BF5		CALL LiesEingänge	
0x2E	7F6		SUBI #F,DELTA_E	nur einmal Nachstarten ermöglichen
0x2F	D32		BRC stopR	
0x30	909		CMPI #0,TASTE	Taste gedrückt?
0x31	D11		BRC stopL	Dann linke Walze nochmal starten
0x32	BF2	stopR	CALL HoleZahl	Rechte Walze ermitteln
0x33	1F0		MOVI #F,RO	König
0x34	1D1		MOVI #D,RU	2 DM
0x35	989		CMPI #8,ZAHL09	
0x36	D48		BRC R_fertig	
0x37	1E0		MOVI #E,RO	Krone
0x38	191		MOVI #9,RU	1 DM
0x39	979		CMPI #7,ZAHL09	
0x3A	D48		BRC R_fertig	
0x3B	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3C	969		CMPI #6,ZAHL09	
0x3D	D48		BRC R_fertig	
0x3E	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x3F	959		CMPI #5,ZAHL09	
0x40	D48		BRC R_fertig	
0x41	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x42	939		CMPI #3,ZAHL09	
0x43	D48		BRC R_fertig	
0x44	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x45	919		CMPI #1,ZAHL09	
0x46	D48		BRC R_fertig	
0x47	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x48	BEB	R_fertig	CALL Warteschleife	
0x49	BF2	stopM	CALL HoleZahl	Mittlere Walze ermitteln
0x4A	1F2		MOVI #F,MI	König
0x4B	989		CMPI #8,ZAHL09	
0x4C	D5A		BRC M_fertig	
0x4D	1C2		MOVI #C,MI	1,60 DM
0x4E	979		CMPI #7,ZAHL09	
0x4F	D5A		BRC M_fertig	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comments
				Assemble Instructions
0x50	142		MOVI #4,MI	40 Pf (2x)
0x51	959		CMPI #5,ZAHL09	
0x52	D5A		BRC M_fertig	
0x53	182		MOVI #8,MI	80 Pf (2x)
0x54	939		CMPI #3,ZAHL09	
0x55	D5A		BRC M_fertig	
0x56	1A2		MOVI #A,MI	1,20 DM (2x)
0x57	919		CMPI #1,ZAHL09	
0x58	D5A		BRC M_fertig	
0x59	1D2		MOVI #D,MI	2 DM (2x)
0x5A	BEB	M_fertig	CALL Warteschleife	
				Gewinnabtastung
				GEWINNSYMBOL wurde in HoleZahl genullt
				CHANCE1 wurde in Warteschleife genullt
0x5B	109		MOVI #0,CHANCE2	
0x5C	BF8		CALL LöscheDelta	
0x5D	840	ChanceA	CMP LO,RO	
0x5E	E62		BRZ erste_Chance	
0x5F	841		CMP LO,RU	
0x60	E62		BRZ erste_Chance	
0x61	C63		GOTO ChanceB	
0x62	048	erste_Chance	MOV LO,CHANCE1	
0x63	830	ChanceB	CMP LU,RO	
0x64	E68		BRZ zweite_Chance	
0x65	831		CMP LU,RU	
0x66	E68		BRZ zweite_Chance	
0x67	C69		GOTO Mitte	
0x68	039	zweite_Chance	MOV LU,CHANCE2	
0x69	9E2	Mitte	CMPI #E,MI	König (Joker) in der Mitte?
0x6A	D74		BRC Joker_ja	
0x6B	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x6C	E6E		BRZ gewinn1	
0x6D	C6F		GOTO test2	
0x6E	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x6F	829	test2	CMP MI,CHANCE2	
0x70	E72		BRZ gewinn2	
0x71	C7B		GOTO symbol	
0x72	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x73	C7B		GOTO symbol	
0x74	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x75	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x76	D78		BRC aaa	
0x77	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x78	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x79	D7B		BRC symbol	
0x7A	09B		MOV CHANCE2,GEWINNSYMBOL	bester Gewinn
0x7B	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x7C	E7F		BRZ G100	
0x7D	D80		BRC Gewinn120	
0x7E	C8F		GOTO GewinnBis80Pf	
0x7F	1A5	G100	MOVI #A,DELTA_D	Hex 10
0x80	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x81	E83		BRZ G120	
0x82	C84		GOTO Gewinn140	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x83	1C5	G120	MOVI #C,DELTA_D	Hex 12
0x84	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x85	E87		BRZ G140	
0x86	C88		GOTO Gewinn160	
0x87	1E5	G140	MOVI #E,DELTA_D	Hex 14
0x88	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x89	E8C		BRZ G160	
0x8A	D99		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x8B	C90		GOTO SoSpTest	
0x8C	105	G160	MOVI #0,DELTA_D	Hex 16
0x8D	116		MOVI #1,DELTA_E	Hex 16
0x8E	C90		GOTO SoSpTest	
0x8F	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x90	90A	SoSpTest	CMPI #0,STATUS	Laufen gerade keine Sonderspiele?
0x91	E9B		BRZ GongTest	
0x92	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x93	D99		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x94	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x95	E99		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x96	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x97	D99		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x98	CA2		GOTO GewinnAnzeige	
0x99	145	Erhöhung	MOVI #4,DELTA_D	2 DM (Hex 20)
0x9A	116		MOVI #1,DELTA_E	
0x9B	906	GongTest	CMPI #0,DELTA_E	Nichts gewonnen?
0x9C	D9F		BRC Ton	
0x9D	905		CMPI #0,DELTA_D	
0x9E	EA2		BRZ GewinnAnzeige	Dann nur anzeigen
0x9F	111	Ton	MOVI #1,RU	
0xA0	FE1		DOT RU	Sound erzeugen, Ausgang 1 auf 1 setzen
0xA1	FEA		DOT STATUS	
0xA2	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA3	F3C		DISP #3,GELD_PF_EINER	
0xA4	BEB		CALL Warteschleife	setzt auch RO auf 0 wg. EXRA
0xA5	F02		DISOUT	
0xA6	F0F		EXRA	
0xA7	BE4		CALL DeltaAdd	
0xA8	BF8		CALL LöscheDelta	Geldgewinne löschen
0xA9	90A		CMPI #0,STATUS	Laufen keine Sonderspiele?
0xAA	EC3		BRZ SoGewinn	dann EFE oder FFF testen
0xAB	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xAC	1F5		MOVI #F,DELTA_D	vorher -1 Sonderspiel (akt. Spiel)
0xAD	BE2		CALL DeltaSub	
0xAE	BF8		CALL LöscheDelta	
0xAF	7F2		SUBI #F,MI	Tableau testen: König in der Mitte?
0xB0	DC2		BRC EndeTableau	kein König in der Mitte

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xB1	90E	Tableau	CMPI #0,S_ZEHNER	Pos > 16?
0xB2	DC2		BRC EndeTableau	dann keine Tableaugewinne
0xB3	99D		CMPI #9,S_EINER	Pos > 10?
0xB4	DBC		BRC Pos16	dann Pos16 und Pos15 prüfen
0xB5	EBA		BRZ Gewinn10S	Pos = 10? dann 10 Sonderspiele
0xB6	125		MOVI #2,DELTA_D	Pos < 10? dann mind. 2 Sonderspiele
0xB7	90D		CMPI #0,S_EINER	Pos = 1?
0xB8	DC2		BRC EndeTableau	
0xB9	585	Gewinn20S	ADDI #8,DELTA_D	20 Sonderspiele geben:
				2 Sonderspiele aufstocken auf 10
				und zusätzlich...
0xBA	5A5	Gewinn10S	ADDI #A,DELTA_D	10 Sonderspiele geben
0xBB	FB6		ADC DELTA_E	
0xBC	9ED	Pos16	CMPI #E,S_EINER	Pos > 15?
0xBD	DC0		BRC Gewinn50S	Dann 50 Sonderspiele
0xBE	EC1		BRZ Gewinn2S	Pos = 15? dann 2 Sonderspiele
0xBF	CC2		GOTO EndeTableau	
0xC0	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele (Hex 32)
0xC1	125	Gewinn2S	MOVI #2,DELTA_D	
0xC2	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xC3	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >15 Sonderspiele gewonnen?
0xC4	DCF		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xC5	9A5		CMPI #A,DELTA_D	Genau 10 gewonnen?
0xC6	ECF		BRZ SoAnzeige	dann anzeigen
0xC7	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xC8	DCD		BRC ZehnS	
0xC9	ECB		BRZ FünfS	
0xCA	CCF		GOTO SoAnzeige	
0xCB	155	FünfS	MOVI #5,DELTA_D	5 Sonderspiele geben
0xCC	CCE		GOTO SoGewonnen	
0xCD	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben
0xCE	12A	SoGewonnen	MOVI #2,STATUS	Statusregister auf "Sonderspiele" setzen
0xCF	FEA	SoAnzeige	DOT STATUS	Keine Sonderspiele?
0xD0	EDF		BRZ Aufräumen	dann keine Anzeige und aufräumen
0xD1	F0F		EXRA	Sonderspiele-Gewinn anzeigen
0xD2	F2D		DISP #2,S_EINER	
0xD3	BEB		CALL Warteschleife	
0xD4	F0F		EXRA	
0xD5	F02		DISOUT	
0xD6	F0E		EXRM	Sonderspiele-Register einblenden
0xD7	BE4		CALL DeltaAdd	mögl. Sonderspiele-Gewinn addieren
0xD8	90E		CMPI #0,S_ZEHNER	Sonderspiele-Zähler 0?
0xD9	DDE		BRC SoAnzeigeEnde	
0xDA	90D		CMPI #0,S_EINER	
0xDB	DDE		BRC SoAnzeigeEnde	
0xDC	10A		MOVI #0,STATUS	Dann Status zurücksetzen
0xDD	F7A		MAS STATUS	und in Arbeitsregister kopieren
0xDE	F0E	SoAnzeigeEnde	EXRM	Sonderspiele-Register ausblenden
0xDF	102	Aufräumen	MOVI #0,MI	Walzen nullen

[illegible]